ReflexiveHull

Let R be an arbitrary ring. For a left[1] R-module M the evaluation map $M \xrightarrow{\varepsilon} M^{**}$ is natural, where ${}^* = \mathrm{Hom}_R(-,R)$ is the dualizing functor. M is called *reflexive* if the evaluation map $M \xrightarrow{\varepsilon} M^{**}$ is an isomorphism. So in general one is intereseted in the kernel and cokernel of ε.

Now let $P_1 \xrightarrow{\varphi} P_0 \to M \to 0$ be an exact sequence with P_1 and P_0 finitely generated projective modules. We call such a sequence a *finite projective presentation* of M. Following [Aus66] we define for M its so-called Auslander dual module $M^\top := \mathrm{coker}(P_0^* \xrightarrow{\varphi^*} P_1^*)$. The isomorphism class of $M^\top$ depends on the presentation of M, but its projective equivalence class depends only on (the isomorphism class of) M (cf. [AB69] and [Eis95, Exercise A3.22]). Further, the so-called evaluation sequence

$$0 \to \mathrm{Ext}^1_R(M^\top, R) \to M \xrightarrow{\varepsilon} M^{**} \to \mathrm{Ext}^2_R(M^\top, R) \to 0$$

is exact (cf. [Aus66] and [HS97, Exercise IV.7.3]). Hence: $\ker(\varepsilon) = \mathrm{Ext}^1_R(M^\top, R)$ and $\mathrm{coker}(\varepsilon) = \mathrm{Ext}^2_R(M^\top, R)$.

In [AB69] a module M with a finite projective presentation is called k-*torsion* free if

$$\mathrm{Ext}^i_R(M^\top, R) = 0 \text{ for } 1 \leq i \leq k.$$

Hence, a module M with a finite projective presentation is reflexive, iff M is 2-torsion free.

For rings where

(Refl) for a finitely generated module M its double-dual M^{**} is reflexive,

one may call M^{**} the *reflexive hull* of M. For commutative noetherian rings this is the case (for a noncommutative two-side noetherian ring see [AB69, the Propositions on p. 143f and the Remark on p. 145]: we need a ring where whenever $\mathrm{Ext}^2_R(N, R)$ is projective for a finitely generated module N, then $\mathrm{Ext}^2_R(N, R) = 0$).

From now on let R be a left Ore (not necessarily commutative) domain (e.g. a left noetherian integral domain). It now makes sense to speak about the torsion submodule $\mathrm{t}(M)$ of the left module M. For a module M with a finite projective presentation there is a natural isomorphism $\mathrm{t}(M) = \mathrm{Ext}^1_R(M^\top, R)$ (cf. [AB69, Pal70, Kas95] and [Qua99] for Ore domains[2]). Hence, torsion freeness is equivalent to 1-torsion freeness. On the other hand, for a finitely generated R-module M, torsion freeness is equivalent to M being a submodule of some free R-module of finite rank.

We consider a torsion free module M, which we can think of as embedded in the free module $M \overset{\iota}{\hookrightarrow} F := R^a$ for some finite a. The diagram

$$\begin{array}{ccc} M & \xrightarrow{\iota} & F \\ {\scriptstyle \varepsilon_M}\downarrow & & \| {\scriptstyle \varepsilon_F} \\ M^{**} & \xrightarrow{\iota^{**}} & F^{**} \end{array}$$

enables one to also embed M^{**} in $F = F^{**}$. In case R additionally satisfies (Refl) then by this one can compute the reflexive hull of M in R^a: $M \leq M^{\mathrm{refl}} = M^{**} \leq R^a$.

```
> restart;
> with(Involutive): with(homalg):
```

Specify the `homalg`-table of the ring package `Involutive`:

```
> RPI:=`Involutive/homalg`;
```

$$RPI := Involutive/homalg$$

---

[1] Unless stated otherwise, we always consider left modules.

[2] The proof in [Qua99] is reproduced in [PQ03, Corollary 2, p. 12], which is available online.

Use the ring package `Involutive` as the default ring package:

```
> `homalg/default`:=RPI;
```

$$homalg/default := Involutive/homalg$$

Define the ring $R = \mathbb{Q}[x, y, z]$:

```
> var:=[x,y,z];
```

$$var := [x, y, z]$$

The presentation matrix `M`:

```
> MM:=[[0,0,0,y-1,y*z], [x,x^2+y,x*y+y^2+z,2*x*z+y*z-y+1,x^2+x*y]];
```

$$MM := [[0, 0, 0, y-1, y\,z], [x, x^2+y, x\,y+y^2+z, 2\,x\,z+y\,z-y+1, x^2+x\,y]]$$

The modules $M := \text{coker}\left(R^{1\times 2} \xrightarrow{\texttt{M}} R^{1\times 5}\right)$, i.e. it is a module on 5 generators with 2 relations:

```
> M:=Cokernel(MM,var);
```

$$\begin{aligned}M := [&[[1, 0, 0, 0, 0] = [1, 0, 0, 0, 0], [0, 1, 0, 0, 0] = [0, 1, 0, 0, 0],\\ &[0, 0, 1, 0, 0] = [0, 0, 1, 0, 0], [0, 0, 0, 1, 0] = [0, 0, 0, 1, 0],\\ &[0, 0, 0, 0, 1] = [0, 0, 0, 0, 1]],\\ &[[0, 0, 0, y-1, y\,z], [x, x^2+y, x\,y+y^2+z, 2\,x\,z+y\,z-y+1, x^2+x\,y]],\\ &\text{“Presentation”}, 5+15\,s+s^2\,(\frac{15}{1-s}+\frac{10}{(1-s)^2}+\frac{3}{(1-s)^3}), [15, 10, 3]]\end{aligned}$$

The module M is torsion free:

```
> IsTorsionFree(M,var);
```

$$true$$

But not reflexive:

```
> IsReflexive(M,var);
```

$$false$$

Compute the double-dual M^{**} (although we don't need this computation in what follows). It is a module on 4 generators with 1 relation:

```
> HHM:=HomHom_R(M,var);
```

$$HHM := \left[\left[\left[1, 0, 0, 0\right] = \begin{bmatrix}0\\0\\0\\1\end{bmatrix}, [0, 1, 0, 0] = \begin{bmatrix}0\\x\,y+y^2+z\\x\\0\end{bmatrix}, [0, 0, 1, 0] = \begin{bmatrix}x\,y+y^2+z\\0\\x^2+y\\0\end{bmatrix},\right.\right.$$

$$\left.[0, 0, 0, 1] = \begin{bmatrix}-x\\x^2+y\\0\\0\end{bmatrix}\right], [[0, x^2+y, -x, -x\,y-z-y^2]], \text{“Presentation”},$$

$$\left.4+12\,s+s^2\,(\frac{12}{1-s}+\frac{8}{(1-s)^2}+\frac{3}{(1-s)^3}), [12, 8, 3]\right]$$

M^{**} is reflexive (as expected):

```
> IsReflexive(HHM,var);
```

$$true$$

But M^{**} is not stably free and hence not free:

```
> IsStablyFreeFFR(HHM,var);
```

$$false$$

Embed the torsion free M into a free module $M \overset{\iota}{\hookrightarrow} F = R^{1\times 3}$:

```
> iota:=ParametrizeModule(M,var);
```

$$\iota := \begin{bmatrix} x\,y+y^2+z & 0 & z^2\,y^2-y^2\,z+2\,z^2\,y+z^3-x\,z-z^2+x+y+z \\ 0 & -x\,y-z-y^2 & z-y\,z+z^2\,y \\ -x & x^2+y & x\,z-x\,z^2-x \\ 0 & 0 & -y\,z \\ 0 & 0 & y-1 \end{bmatrix}$$

The free module F (the target of ι):

```
>  F:=FreeHullModule(iota,var);
```

$$F := [[[1,\,0,\,0] = [1,\,0,\,0],\ [0,\,1,\,0] = [0,\,1,\,0],\ [0,\,0,\,1] = [0,\,0,\,1]],\ [[0,\,0,\,0]],\ \text{“Presentation”},\ \frac{3}{(1-s)^3},\ [0,\,0,\,3]]$$

ι is indeed an embedding:

```
>  IsHom(M,iota,F,var);
```

$$\mathit{true}$$

```
>  IsInjective(M,iota,F,var);
```

$$\mathit{true}$$

The image of M in F again given by a finite presentation[3]:

```
>  Image(iota,F,var,"USE_IMAGE_OF_GENERATORS");
```

$$\begin{aligned}&[[[1,\,0,\,0,\,0,\,0] = [x\,y+y^2+z,\,0,\,z^2\,y^2-y^2\,z+2\,z^2\,y+z^3-x\,z-z^2+x+y+z],\\ &[0,\,1,\,0,\,0,\,0] = [0,\,-x\,y-z-y^2,\,z-y\,z+z^2\,y],\\ &[0,\,0,\,1,\,0,\,0] = [-x,\,x^2+y,\,x\,z-x\,z^2-x],\ [0,\,0,\,0,\,1,\,0] = [0,\,0,\,-y\,z],\\ &[0,\,0,\,0,\,0,\,1] = [0,\,0,\,y-1]],\\ &[[0,\,0,\,0,\,y-1,\,y\,z],\ [x,\,x^2+y,\,x\,y+y^2+z,\,2\,x\,z+y\,z-y+1,\,x^2+x\,y]],\\ &\text{“Presentation”},\ 5+15\,s+s^2\,(\frac{15}{1-s}+\frac{10}{(1-s)^2}+\frac{3}{(1-s)^3}),\ [15,\,10,\,3]]\end{aligned}$$

The induced map $M^{**} \xrightarrow{\mu:=\iota^{**}} F^{**} = F$ (compare μ, which is the embedding of M^{**} into F, with ι, which is the embedding of M into F):

```
>  mu:=HomHomMap_R(M,iota,F,var);
```

$$\mu := \begin{bmatrix} 0 & 0 & 1 \\ 0 & x\,y+y^2+z & 0 \\ x\,y+y^2+z & 0 & 0 \\ -x & x^2+y & 0 \end{bmatrix}$$

Compute the image of M^{**} inside F, i.e. the reflexive hull of $M \le F$ in F (compare the list of relations with that of M^{**} above):

```
>  HH:=Image(mu,F,var,"USE_IMAGE_OF_GENERATORS");
```

$$\begin{aligned}&\mathit{HH} := [[[1,\,0,\,0,\,0] = [0,\,0,\,1],\ [0,\,1,\,0,\,0] = [0,\,x\,y+y^2+z,\,0],\\ &[0,\,0,\,1,\,0] = [x\,y+y^2+z,\,0,\,0],\ [0,\,0,\,0,\,1] = [-x,\,x^2+y,\,0]],\\ &[[0,\,x^2+y,\,-x,\,-x\,y-z-y^2]],\ \text{“Presentation”},\\ &4+12\,s+s^2\,(\frac{12}{1-s}+\frac{8}{(1-s)^2}+\frac{3}{(1-s)^3}),\ [12,\,8,\,3]]\end{aligned}$$

```
>  IsReflexive(HH,var);
```

$$\mathit{true}$$

[3] Unless invoked with the option `"USE_IMAGE_OF_GENERATORS"`, the procedure `Image` tries to find fewer generators of the image.

Author: MOHAMED BARAKAT

Date:	2007-07-24
Last modified:	2007-07-30 11:41

REFERENCES

[AB69] Maurice Auslander and Mark Bridger, *Stable module theory*, Memoirs of the American Mathematical Society, No. 94, American Mathematical Society, Providence, R.I., 1969. MR MR0269685 (42 #4580) 1

[Aus66] Maurice Auslander, *Coherent functors*, Proc. Conf. Categorical Algebra (La Jolla, Calif., 1965), Springer, New York, 1966, pp. 189–231. MR MR0212070 (35 #2945) 1

[BCG+03] Y. A. Blinkov, C. F. Cid, V. P. Gerdt, W. Plesken, and D. Robertz, *The* MAPLE *Package* JANET*: I. Polynomial Systems. II. Linear Partial Differential Equations*, Proc. 6th Int. Workshop on Computer Algebra in Scientific Computing, Passau, Germany, 2003, (http://wwwb.math.rwth-aachen.de/Janet).

[BR] Mohamed Barakat and Daniel Robertz, homalg – *A meta-package for homological algebra*, submitted. arXiv:math.AC/0701146 and (http://wwwb.math.rwth-aachen.de/homalg).

[BR07] ———, homalg *project*, 2004-2007, (http://wwwb.math.rwth-aachen.de/homalg).

[Eis95] David Eisenbud, *Commutative algebra*, Graduate Texts in Mathematics, vol. 150, Springer-Verlag, New York, 1995, With a view toward algebraic geometry. MR MR1322960 (97a:13001) 1

[HS97] P. J. Hilton and U. Stammbach, *A course in homological algebra*, second ed., Graduate Texts in Mathematics, vol. 4, Springer-Verlag, New York, 1997. MR MR1438546 (97k:18001) 1

[Kas95] Masaki Kashiwara, *Algebraic study of systems of partial differential equations*, Mém. Soc. Math. France (N.S.) (1995), no. 63, xiv+72. MR MR1384226 (97f:32012) 1

[Pal70] V. P. Palamodov, *Linear differential operators with constant coefficients*, Translated from the Russian by A. A. Brown. Die Grundlehren der mathematischen Wissenschaften, Band 168, Springer-Verlag, New York, 1970. MR MR0264197 (41 #8793) 1

[PQ03] Jean-François Pommaret and Alban Quadrat, *A functorial approach to the behaviour of multidimensional control systems*, Int. J. Appl. Math. Comput. Sci. **13** (2003), no. 1, 7–13, (http://www-sop.inria.fr/cafe/Alban.Quadrat/Publications/amcs.pdf) Multidimensional systems nD and iterative learning control (Czocha Castle, 2000),. MR MR1986488 (2004d:93026) 1

[Qua99] Alban Quadrat, *Analyse algébrique des systèmes de contrôle linéaires multidimensionnels*, Ph.D. thesis, Ecole Nationale des Ponts et Chaussées, CERMICS, France, September 1999. 1

LEHRSTUHL B FÜR MATHEMATIK, RWTH-AACHEN UNIVERSITY, 52062 GERMANY
E-mail address: mohamed.barakat@rwth-aachen.de